ALGORITHM 52
A SET OF TEST MATRICES

JOHN R. HERNDON
Stanford Research Institute, Menlo Park, California


```
procedure  TESTMATRIX (n,A);  value n;  integer n;
           real array A;
comment  This procedure places in A an n × n matrix whose
  inverse and eigenvalues are known. The n-th row and the n-th
  column of the inverse are the set: 1, 2, 3, . . . , n. The matrix
  formed by deleting the n-th row and the n-th column of the
  inverse is the identity matrix of order n−1;
begin      integer i, j;
           real t, c, d, f;
           c := t × (t+1) × (t+t−5)/6;
           d := 1/c;
           A[n, n] := −d;
           for i := 1 step 1 until n−1 do
             begin f := i;
               A[i, n] := d × f;
               A[n, i] := A[i, n];
               A[i, i] := d × (c−f × f);
               for j := 1 step 1 until i−1 do
                 begin t := j;
                   A[i, j] := −d × f × t;
                   A[j, i] := A[i, j]
                 end
             end
end TESTMATRIX;
```

CERTIFICATION OF ALGORITHM 52
A SET OF TEST MATRICES (J. R. Herndon, *Comm. ACM*, Apr. 1961)

H. E. GILBERT
University of California at San Diego, La Jolla, Calif.

The statement

```
c := t × (t+1) × (t+t−5)/6;
```

was changed to

```
c := n × (n+1) × (n+n−5)/6;
```

to make the inverse have the form described in the algorithm. The algorithm was translated to FORTRAN and tested with a matrix eigenvalue program on the CDC 1604 computer at UCSD.

The eigenvalues for the 20 × 20 test matrix are:

1.	1.000000
2.	1.000000
⋮	⋮
19.	.01636693
20.	−.02493833

REMARK ON ALGORITHM 52
A SET OF TEST MATRICES (John R. Herndon, *Comm. ACM*, Apr. 1961)

G. H. DUBAY
University of St. Thomas, Houston, Tex.

In the assignment statement

```
c := t×(t + 1)×(t + t − 5)/6;   (a)
```

the t is undefined. A suitable definition would be provided by preceding (a) with t := n;

REMARKS ON AND CERTIFICATION OF ALGORITHM 52
A SET OF TEST MATRICES [J. R. Herndon, *Comm. ACM*, Apr. 1961]

P. NAUR
Regnecentralen, Copenhagen, Denmark

In addition to inserting the correction indicated by H. E. Gilbert [*Comm. ACM* (Aug. 1961), 339] the algorithm was simplified by using the simultaneous assignment and by eliminating the local variables *t* and *f*. The resulting algorithm is as follows:

```
procedure TESTMATRIX(n,A);
value n;  integer n;  real array A;
comment ALG. 52: This procedure places in A an n × n matrix
  whose inverse and eigenvalues are known. The nth row and the
  nth column of the inverse are the set: 1, 2, 3, ..., n. The matrix
  formed by deleting the nth row and the nth column of the in-
  verse is the identity matrix of order n−1;
begin integer i,j;  real c,d;
c := n × (n+1) × (n+n−5)/6;
d := 1/c;
A[n,n] := −d;
for i := 1 step 1 until n−1 do
  begin
  A[i,n] := A[n,i] := d × i;
  A[i,i] := d × (c−i↑2);
  for j := 1 step 1 until i−1 do A[i,j] := A[j,i] := −d × i × j
  end
end TESTMATRIX;
```

This version of the algorithm was successfully run in the GIER ALGOL system together with the inversion procedures INVERSION II and gjr (see Certification of Algorithm 120 below). From the figures produced by INVERSION II it looks as if the determinant of these matrices is given by $6/(n(n+1)(5-2n))$, which is also the value of the element $A[n,n]$. For $n > 3$ the absolutely greatest element is $A[1,1] = 1 + A[n,n]$.

CERTIFICATION OF ALGORITHM 52
A SET OF TEST MATRICES [J. R. Herndon, *Comm. ACM*, Apr. 1961]

J. S. HILLMORE
Elliott Bros. (London) Ltd., Borehamwood, Herts., England

The algorithm was corrected as recommended by H. E. Gilbert in his certification [*Comm. ACM*, Aug. 1961] and then successfully run using the Elliott ALGOL translator on the National-Elliott 803. The matrices so generated were used to test the matrix inversion procedure GJR given by H. R. Schwarz in his article "An Introduction to ALGOL" [Comm. ACM, Feb. 1962].

ADDITIONAL REMARKS ON ALGORITHM 52 A SET OF TEST MATRICES [J. R. Herndon, *Comm. ACM* (Apr. 1961), 180]

P. NAUR
Regnecentralen, Copenhagen, Denmark

From an inspection of the results of eigenvalue-finding algorithms I conclude that all but two of the eigenvalues of TESTMATRIX are unity while the two remaining are given by the expressions $6/(p\times(n+1))$ and $p/(n\times(5-2\times n))$ where

$$p = 3 + \text{sqrt}\ ((4\times n-3) \times (n-1) \times 3/(n+1)).$$

These expressions have been used for the determination of absolute errors of the eigenvalues calculated by JACOBI, Algorithm 85, and Householder Tridiagonalisation, etc. as reported below. They were also used to calculate the following table (using GIER ALGOL, with 29 significant bits):

n	*Determinant*	*Eigenvalues Differing from unity*	
3	−.500 000 00	.224 744 87	−2.224 744 9
4	−.100 000 00	.153 112 89	−.653 112 89
5	−.040 000 000	.113 238 08	−.353 238 08
6	−.020 408 163	.088 290 570	−.231 147 71
7	−.011 904 762	.071 428 571	−.166 666 67
8	−.007 575 757 6	.059 386 081	−.127 567 90
9	−.005 128 205 2	.050 422 549	−.101 704 60
10	−.003 636 363 6	.043 532 383	−.083 532 383
11	−.002 673 796 8	.038 097 478	−.070 183 039
12	−.002 024 291 5	.033 718 770	−.060 034 559
13	−.001 569 858 7	.030 128 103	−.052 106 125
14	−.001 242 236 0	.027 139 206	−.045 772 747
15	−.001 000 000 0	.024 619 013	−.040 619 013
16	−.000 816 993 47	.022 470 157	−.036 359 046
17	−.000 676 132 52	.020 619 902	−.032 790 288
18	−.000 565 930 96	.019 012 916	−.029 765 605
19	−.000 478 468 90	.017 606 429	−.027 175 807
20	−.000 408 163 27	.016 366 903	−.024 938 332

The figures for $n = 20$ agree very well with the results quoted by H. E. Gilbert in his certification [*Comm. ACM 4* (Aug. 1961), 339].